


NoiseCancellation_OpAmps



```
.OPTIONS GMIN=1e-15 METHOD=gear ABSTOL=1e-15 TEMP=27 srcsteps = 1 gminsteps = 1
VCC VCC 0 DC 5
VEE VEE 0 DC -5
```

```
.include Noise_1m.txt
```

```
q1 v2 v0 vt pnpl 1
q2 v4 v1 vt pnpl 1
q3 v2 v2 vr1 npnv 1
q4 v4 v3 vr2 npnv 1
q5 VCC v5 v6 npnv 1
q6 v7 v6 VEE npnv 1
q7 v9 v9 v8 npnv 1
q8 v7 v7 v8 pnpl 1
q9 VCC v9 v10 npnv 10
q10 VEE v7 v10 pnpl 10
```

```
IT VCC vt 26u
I2 VCC v9 26u
r1 vr1 VEE 50k
r2 vr2 VEE 50k
r3 v6 VEE 60k
c1 v4 v7 15p
```

```
b1 0 v1 v = v(vb1)*v(OUT_1m)
b2 v2 v3 v = v(vb2)*v(OUT_1m)
b3 v4 v5 v = v(vb3)*v(OUT_1m)
b4 v10 v0 v = v(vb4)*v(OUT_1m)
```

```
vb1v vb1 0 DC 0
vb2v vb2 0 DC 0
vb3v vb3 0 DC 0
vb4v vb4 0 DC 0
```

```
.tran 500u 1s 0 500u UIC
```

```
.nodeset v(vcc)=5, v(vee)=-5, v(out_1m)=0.000960589, v(v2)=-4.40659, v(v0)=-5.65798e-05, v(vt)=0.665323,
+ v(v4)=-3.80325, v(v1)=0, v(vr1)=-4.99936, v(v3)=-4.40659, v(vr2)=-4.99935, v(v5)=-3.80325,
+ v(v6)=-4.38927, v(v7)=-0.68254, v(v9)=0.609309, v(v8)=-6.56659e-05, v(v10)=-5.65798e-05, v(vb1)=0,
+ v(vb2)=0, v(vb3)=0, v(vb4)=0
```

```
##0=====See Noise Cancelation Work Inside An OpAmp=====
```

```
.control
```

```
set pensize = 1
```

```
##1=====See Voltage Nodes With Noise Turned Off=====
```

```
run
```

```
nodesets "nodeshere.txt"
```

```
plot v1 ylimit -3m 3m title B1_Off
```

```
plot v0 -v1 ylimit -.3m .3m title B1_Off
```

```
##2=====Input Stage Noise Is Turned ON=====
```

```
alter vb1v dc = 1
```

```
run
```

```
plot v1 ylimit -3m 3m title B1_On
```

```
plot v0 -v1 ylimit -.3m .3m title B1_On
```

```
##3=====Turn Around Noise Is Turned ON=====
```

```

alter vb1v          dc = 0
alter vb2v          dc = 1
run
plot v1             ylimit -3m 3m   title B2_On_R_50K
plot v0 -v1         ylimit -.3m .3m title B2_On_R_50K
plot v0 -v1         ylimit -3m 3m   title B2_On_R_50K

```

***#4=====Degeneration_Resistors_Are_Removed=====**

```

alter r1             resistance = 1
alter r2             resistance = 1
run
plot v0 -v1         ylimit -3m 3m   title B2_On_R_1

```

***#5=====Second_Stage_Noise_Is_Turned_ON=====**

```

alter r1             resistance = 50k
alter r2             resistance = 50k
alter vb2v           dc = 0
alter vb3v           dc = 1
run
plot v0 -v1         ylimit -.3m .3m title B3_On
plot v4 v5

```

***#6=====Output_Stage_Noise_Is_Turned_ON=====**

```

alter vb3v           dc = 0
alter vb4v           dc = 1
run
plot v10 v0 -v1     ylimit -3m 3m   title B4_On

```

.endc

```

.model npnv npn ( VAF=200 IKF=6e-03
+IS=1E-15 NF=1.0 BF=150
+CJE=1.7E-13 CJC=1.8E-13 CJS=2.9E-12 TF=1.6E-9
+XTB=2 EG=1.11 XTI=10 TNOM=25 )

.model npnl pnp ( VAF=200 IKF=0.002
+IS=7.91E-17 NF=1.01 BF=150
+CJE=1.7E-13 CJC=1.8E-13 CJS=2.9E-12 TF=1.6E-9
+XTB=2 EG=1.11 XTI=7 TNOM=25 )

.end

```


=====END_OF_SPICE=====

***#1=====See_Voltage_Nodes_With_Noise_Turned_Off=====**



```

run
nodesets "nodeshere.txt"
plot v1             ylimit -3m 3m   title B1_Off
plot v0 -v1         ylimit -.3m .3m title B1_Off

```




With the Noise Off there is only Offset




***#2=====Input_Stage_Noise_Is_Turned_ON=====**

```
alter vb1v          dc = 1
run
plot v1            ylimit -3m 3m    title B1_On
plot v0 -v1        ylimit -.3m .3m  title B1_On
```




**With Noise B1 ON, there is only Offset at the inputs.
The output will move to cancel out any noise
at the input.**



```


*#3=====Turn_Around_Noise_Is_Turned_ON=====
alter vb1v          dc = 0
alter vb2v          dc = 1
run
plot v1             ylimit -3m 3m      title B2_On_R_50K
plot v0 -v1         ylimit -.3m .3m     title B2_On_R_50K
plot v0 -v1         ylimit -3m 3m      title B2_On_R_50K

```




With Noise B2 ON, there is a little at the inputs. The output will create enough voltage across the input to cancel out B2.

The degeneration resistors R1 and R2 are effectively providing Input stage gain.




*#4=====Degeneration_Resistors_Are_Removed=====

```
alter r1 resistance = 1
alter r2 resistance = 1
run
plot v0 -v1 ylimit -3m 3m title B2_On_R_1
```



With the degeneration resistors R1 and R2 removed, there is no effectively gain.

The output will now have to create a large enough voltage across the input to cancel out B2.




*#5=====Second_Stage_Noise_Is_Turned_ON=====

```

alter r1 resistance = 50k
alter r2 resistance = 50k
alter vb2v dc = 0
alter vb3v dc = 1
run
plot v0 -v1 ylimit -.3m .3m title B3_On
plot v4 v5 title B3_On


```




With Noise B3 ON, there is very little at the inputs. The output will create enough voltage across the input to cancel out B3.

Gain to the noise source now includes the TurnAround stage gain.

Graph 221 - tran203: B3_On



Graph 178 - tran179: B3_On




Notice that the voltage at Node V4

is being moved at the exact amount to cancel out any Noise that appears at node V5.

```


*#6=====Output_Stage_Noise_Is_Turned_ON=====
alter vb3v          dc = 0
alter vb4v          dc = 1
run
plot v10 v0 -v1 ylimit -3m 3m title B4_On

```



With Noise B4 ON, there is very little anywhere. Gain to the noise source now includes the TurnAround stage gain and the second stage.

Very little voltage is needed across the inputs to cancel out the noise of B4.



DC input offset voltage works exactly the same way. A 1mV_RMS noise source equals 1mV_SD of offset. Offset can be thought of as DC noise. Rather than vary as a function of steps in time, it varies over individual units.